

1. Mô hình V-VTEAM

Nguyên nhân ra đời của V-VTEAM ¹ (Volatile Voltage ThrEshold Adaptive Memristor model):

- Các mô hình cũ quá chuyên biệt và nặng tính vật lý
- Nhu cầu thực tế: cần một mô hình đơn giản, tổng quát, mô phỏng nhanh
- Mô hình mới đơn giản mà vẫn chính xác

Trạng thái của Memristor trong mô hình V-VTEAM được mô tả thông qua biến **x** bị chặn trong đoạn [**x_{on}** , **x_{off}**]; là một đại lượng động, thay đổi theo thời gian, thể hiện mức độ "kích hoạt" hay "trạng thái dẫn điện". Đạo hàm của **x** được mô tả theo công thức:

$$\frac{dx}{dt} = \begin{cases} \frac{-x \cdot \beta \cdot (\frac{t}{\tau})^{\beta-1}}{\tau}, & v(t) \leq v_h \\ 0, & v_h < v(t) < v_{th} \\ k \cdot \left(\frac{v(t)}{v_{th}} - 1\right)^\alpha, & v(t) \geq v_{th} \end{cases}$$

Trong đó:

STT	KÝ HIỆU	Ý NGHĨA	STT	KÝ HIỆU	Ý NGHĨA
1	α	Chỉ số tăng trưởng (growth exponent)	5	V_h	Điện áp giữ (hold voltage)
2	β	Chỉ số suy giảm (decay exponent)	6	V_{th}	Ngưỡng kích hoạt (threshold voltage)
3	τ	Hằng số thời gian suy giảm (retention time parameter)	7	$V(t)$	Điện áp đặt lên Memristor (P-N)
4	k	Hằng số tốc độ tăng (growth rate constant)	8	t	Thời gian suy giảm

Mô hình dòng điện-điện áp (I-V) của Memristor trong mô hình V-VTEAM được mô tả thông qua biểu thức tuyến tính bên dưới:

$$i(t) = [R_{ON} + \frac{R_{OFF}-R_{ON}}{x_{off}-x_{on}} \cdot (x(t) - x_{on})]^{-1} \cdot v(t)$$

Trong đó:


- R_{ON} : Giá trị điện trở ở trạng thái LRS.
- R_{OFF} : Giá trị điện trở ở trạng thái HRS.

Chọn $I_{max} \gg \max(I(t))$. sử dụng hàm $\tanh(x)$ để tránh được các lỗi liên quan đến đạo hàm.

$$b_1(x) = x_{max} \tanh\left(\frac{x}{x_{max}}\right)$$

¹ T. Patni, R. Daniels, and S. Kvatinsky, "V-VTEAM: A Compact Behavioral Model for Volatile Memristors," 2024 International Flexible Electronics Technology Conference (IFETC), 2024, pp. 1–4, doi: 10.1109/IFETC61155.2024.10771870.

Hình bên dưới cho thấy hiệu quả trong việc giới hạn dòng điện.



a. $I_{\max}=2\max(I(t))$,

b. $I_{\max}=0.25\max(I(t))$

Hình . (a, b) Mức độ hiệu quả trong việc giới hạn của hai hàm b_0 và b_1 .

Nhận xét: Do đó, sử dụng hàm $\tanh(x)$ sẽ đạt hiệu quả giới hạn dòng điện tốt hơn.

Lúc này, mô hình dòng điện I-V được mô tả như sau:

$$i(t) = I_{\max} \tanh \left(\frac{\left(R_{ON} + \frac{R_{OFF} - R_{ON}}{x_{OFF} - x_{ON}} \cdot (x(t) - x_{ON}) \right)^{-1} \cdot v(t)}{I_{\max}} \right)$$

2. Mô hình V-VTEAM sử dụng VerilogA

Mặc dù có các hàm được dựng sẵn cho việc mô tả quan hệ (idt, ddt) nhưng các hàm này chỉ giành cho các discipline, các biến (VD: real) không thể sử dụng các hàm này.



Hình . Không thể sử dụng $ddt(x)$ để mô tả dx .

Mặt khác, do $V(t)$ không được biết trước nên việc tính toán họ nguyên hàm của dx là bất khả thi. Do đó để sử dụng biến dt để tính đạo hàm dựa trên xấp xỉ Euler là lựa chọn thay thế duy nhất và đơn giản.

Tham số dt là tham số bước nhảy thời gian cho việc xấp xỉ nguyên hàm dựa theo công thức của Euler, để kết quả mô phỏng được chính xác, giá trị dt phải nhỏ hơn chu kỳ nhỏ nhất của nguồn 10^3 lần.

Các tham số khác của mô hình dựa theo Bảng 01. Khớp các tham số của mô hình ².

² T. Patni, R. Daniels, and S. Kvatinsky, "V-VTEAM: A Compact Behavioral Model for Volatile Memristors," 2024 International Flexible Electronics Technology Conference (IFETC), 2024, pp. 1–4, doi: 10.1109/IFETC61155.2024.10771870.

Parameter	Testing Value	Fitting Value
x_{on}	1	
x_{off}	0	
V_{th} [V]	1.8	1.744
V_h [V]	1.4	1.5726
R_{on} [Ω]	30k	
R_{off} [Ω]	15G	
k [m/s]	150	650
α	0.45	0.09999
τ [s]	10m	1.0444
β	5	2.14262

Mã nguồn mô hình V-VTEAM được viết bằng Verilog-A:

```

`include "constants.vams"
`include "disciplines.vams"

/// Define a new unit for x-state
nature state
    access = stat;
    units = "x1";
    abstol = 1u;
endnature

discipline state_disc
    potential state;
enddiscipline

/// Manual integration of state variable
module memristor(p, n, xnode);
    inout p, n;
    electrical p, n;

    output xnode;
    state_disc xnode;

    /// Parameters for the VVTEAM memristor model
    parameter real Xon  = 1;
    parameter real Xoff = 0;
    parameter real Xinit = 0.5;
    parameter real Vth  = 1.744;
    parameter real Vh   = 1.5726;
    parameter real Ron  = 30k;
    parameter real Roff = 15G;
    parameter real k    = 650;
    parameter real alpha = 0.099999;
    parameter real tau  = 1.0444;
    parameter real beta  = 2.14262;
    parameter real Imax  = 11u;

```

```

/// User must specify dt same as max step size in
/// transient analysis & must be at least 3 orders
/// smaller than T period of the source
/// dt = 0, auto compute via t_last_compute;
parameter real dt = 0;

/// Approx Zero
parameter approx_zero = 15a;

/// Local variables
real x, dxdt, t_decay, equiv_res, l_calc;
real delta_t, t_last_compute;

analog function real soft_bounded;
  input __value, __max_absolute;
  real __value, __max_absolute;
  begin
    soft_bounded = __max_absolute * tanh(__value / __max_absolute);
  end
endfunction

analog function real bounded;
  input __value, __lower, __upper;
  real __value, __lower, __upper;
  if (__value < __lower)
    bounded = __lower;
  else if (__value > __upper)
    bounded = __upper;
  else
    bounded = __value;
endfunction

analog function real dxdt_decay;
  input __t;  input __x;
  real __t;  real __x;

  begin
    if (__t > approx_zero)
      dxdt_decay = (-1 * __x * beta * pow(__t / tau, beta - 1) / tau);
    else
      dxdt_decay = 0;
    end
endfunction

analog function real dxdt_growth;
  input __Vpn; real __Vpn;
  real __base;
  begin

```

```

    __base = (__Vpn / Vth) - 1;
    if(__base > approx_zero)
        dxdt_growth = k * pow(__base, alpha);
    else
        dxdt_growth = 0;
    end
endfunction

analog begin
    /// Initialize state at simulation start
    @(initial_step) begin
        x = Xinit;
        t_decay = 0;
        delta_t = 0;
        t_last_compute = $abstime;
    end

    /// Update delta_t
    if( dt < approx_zero ) begin
        /// Compute delta_t
        delta_t = $abstime - t_last_compute;
        /// Saved current time
        t_last_compute = $abstime;
    end else begin
        delta_t = dt;
    end

    /// Update dx/dt based on input voltage
    if (V(p, n) < Vh) begin
        t_decay = t_decay + delta_t;
        dxdt = dxdt_decay(t_decay, x);
    end else begin
        t_decay = 0;
        if (V(p, n) > Vth) begin
            dxdt = dxdt_growth(V(p, n));
            t_decay = 0;
        end else begin
            dxdt = 0;
        end
    end

    /// Manually integrate to update x
    x = bounded(x + dxdt * delta_t, Xoff, Xon);

    /// Update resistance
    eqiv_res = (Ron + (((Roff - Ron) / (Xoff - Xon)) * (x - Xon)));

    /// Compute current and update current
    I(p,n) <+ soft_bounded(V(p, n) / eqiv_res, Imax);


```

```


/// Update xnode
stat(xnode) <+ x;
end
endmodule

```


3. Thiết lập mô phỏng



Hình . Sơ đồ mô phỏng.




Hình . Thiết lập mô trường.




Hình . Thiết lập chế độ đánh giá.

(Lưu ý rằng bước mô phỏng phải bằng với bước thiết lập trong mô hình Memristor.)



Hình . Thiết lập ADE.

4. Kết quả mô phỏng



Hình . Mô hình được hiệu chỉnh theo một mô hình memristor bay hơi Ag/SiOx/C/W dựa trên drift/diffusion của ion ³.


³ W. Wang et al., "Volatile Resistive Switching Memory Based on Ag Ion Drift/Diffusion—Part II: Compact Modeling," IEEE Transactions on Electron Devices, Vol. 66, No. 9, pp. 3802-3808, September 2019.




Hình . Dạng sóng dòng điện - điện áp và trạng thái với các giá trị τ khác nhau.

5. Khớp mô hình thành VTEAM

Hiệu chỉnh các tham số của mô hình V-VTEAM để dạng sóng I-V như mô hình VTEAM (Hình . Dạng sóng mô hình VTEAM ở $F_{\sin}=50\text{MHz}$; $VDD=1\text{V}$. và Hình . Dạng sóng biến trạng thái nội tại của Memristor với w - VTEAM; x_{node} - VVTEAM.).




Hình . Dạng sóng mô hình VTEAM ở $F_{\sin}=50\text{MHz}$; $VDD=1\text{V}$.




Hình . Dạng sóng biến trạng thái nội tại của Memristor với w - VTEAM; x_{node} - VVTEAM.

Pha suy giảm (tăng điện trở) trong mô hình V-VTEAM là hàm độc lập với điện áp, chỉ phụ thuộc vào thời gian kể từ lúc điện áp P-N dưới ngưỡng giữ (V_{hold}). Trong khi đó, pha suy giảm mô hình VTEAM là hàm phụ thuộc vào điện áp. Để mô phỏng mô hình VTEAM, phải "giảm thời gian suy giảm" thông qua τ . Do đó, tùy vào tần số hoạt động mà thiết lập các giá trị τ khác nhau. Hình bên dưới là bộ tham số cho mô hình VTEAM hoạt động ở $f_{req}=50\text{MHz}$, $V_{DD}=1\text{V}$.



Hình . Tham số mô phỏng mô hình VTEAM, $f=50\text{MHz}$, $V_{DD}=1\text{V}$.



a. V1-V2: "tăng tương"; V2-V3: "giữ"; V3-...: "suy giảm".

b. Đồ thị I-V mô hình VTEAM.

Hình . Dạng sóng I-V mô phỏng VTEAM.

